

DESIGN NOTES

Number 28 in a series from Linear Technology Corporation

November, 1989

A SPICE Op Amp Macromodel for the LT1012

Walt Jung

Introduction

The Boyle, et al.¹, SPICE macromodel for op amps has proven to be quite useful for fast and efficient computer-based IC circuit analysis, used within its limitations. Critics of this type of model point out that it is not optimum for precise transient analysis of amplifiers using complex compensation. On the other hand, the Boyle macromodel may have little match in terms of the computational speed and performance it can achieve, plus how quickly it can be implemented. These virtues are particularly true for lower frequency op amps, or where DC performance parameters are more important.

The Boyle model can be set up to give realistic and quite reasonable working approximations to a variety of IC op amps which use various types of differential transconductance pair front ends. Two fundamental advantages of this model are the relative simplicity and the simulation speed (particularly when a minimum number of junctions are used). Further, the prudent use of the appropriate transistors at the input can simulate real input offset voltage and bias current effects, as well as such IC-unique features as input common-mode clamping², making the overall model much more realistic and akin to real-world ICs. While the original Boyle paper used an NPN bipolar input example (the 741), the topology of the macromodel is also readily adaptable to PNP bipolars as well as JFETs, as design options.

The LT1012

The LT1012 op amp is a popular "universal" high performance internally compensated precision op amp, available in a variety of electrical grades and packages. It uses a

rather unique input stage, comprised of a bias current compensated Super-Beta NPN differential pair. This allows the desirably low drift of an NPN pair to be realized, but with typical bias currents of only 30pA, due to the use of both the Super-Beta process and bias current cancellation. Importantly, the low bias currents are *not* achieved at the expense of poor drift and high voltage noise, as the LT1012 (C grade) accomplishes a $1\mu\text{V}/^\circ\text{C}$ Max. drift, and a $14\text{nV}/\text{Hz}^{1/2}$ Typ. voltage noise.

The LT1012 has recently been broadened in terms of performance grades, with the addition of a premium "LT1012A" grade part, featuring $25\mu\text{V}$ Max. V_{OS} , $0.6\mu\text{V}/^\circ\text{C}$ Max. drift, and $500\mu\text{A}$ Max. supply current. The added "LT1012D" part has a $140\mu\text{V}$ Max. V_{OS} , a $1.7\mu\text{V}/^\circ\text{C}$ Max. drift, and an $800\mu\text{A}$ Max. supply current. All device grades have the unusual combination of performance characteristics which allow use as a low-voltage ($\pm 1.2\text{V}$), low supply current micropower op amp as well as a full $\pm 20\text{V}$ supply range general purpose part. The LT1012 actually exceeds the performance of the industry standard OP-07, doing so at 1/20 the bias/offset currents, and 1/8 of the supply current.

The LT1012 Macromodel

While all of the above may be interesting enough to a designer, how the model imitates the real part is more so. The LT1012 macromodel listed in Figure 1 has a number of features worthy of mention. Note that it is based on the LT1012C room temperature typical specs, taken from the data sheet.

V_{OS} , the input offset voltage of the input pair, is modeled by using two slightly different NPN transistor models, qm1 and qm2. The ratio of their two saturation currents will produce an offset voltage, which is:

$$V_{OS} = KT/q \ln(I_{S1}/I_{S2})$$

With the ratio as shown, this produces the typical $10\mu V$ offset for the LT1012C.

Bias and offset currents are modeled by using a different Bf for the two input pair halves, as:

$$Bf1 = I_{C1}/I_{B1} \text{ and } Bf2 = I_{C2}/I_{B2}$$

The Bf values shown for qm1 and qm2 are those which correspond to $I_B = 30pA$; $I_{OS} = 20pA$. While the gains listed are enormously high (even for Super-Beta transistors) this is not a problem for SPICE, so bias currents in the range of a typical LT1012C are produced.

Other additions to the generic Boyle macromodel are the optional input diode clamps, ddm1 and ddm2, as in the real part (they can be deleted, if not used). The substitution of a current source, I_p , in the place of the R_p of the original model simulates quiescent power supply current. The LT1012, like

many modern day ICs, has a quiescent supply current which is quite constant with supply voltage, thus I_p is more appropriate than a fixed resistor.

The remaining specifications modeled are shown at the head of the listing, consistent with the LT1012C. The model can also be used for the LT1024 (dual LT1012), if the "x" call is added at the end as shown. A sample small signal pulse response waveform of the model is shown in Figure 2, which can be compared to the similar condition scope photo, from the LT1012 data sheet (pg. 7).




Figure 2. Small Signal Transient Response

```
*
* Linear Technology LT1012 op amp model (with calls for LT1024)
* Written: 08-17-1989 10:16:25 Type: Bipolar npn input, internal comp.
* Typical specs:
* Vos=1.0E-05, Iib=3.0E-11, Ios=2.0E-11, GBP=6.0E+05Hz, Phase mar.= 70 deg,
* SR(+)=2.0E-01V/us, SR(-)=1.9E-01V/us, Av= 126 dB, CMRR= 132 dB,
* Vsat(+)= 1 V, Vsat(-)= 1 V, Isc= +/- 12.5 mA, Iq= 380 uA
* (input differential mode clamp active)
*
* Connections: + - V+V-0
.subckt LT1012 3 2 7 4 6
* input
rc1 7 80 8.842E+03
rc2 7 90 8.842E+03
q1 80 2 10 qm1
q2 90 3 11 qm2
ddm1 2 3 ddm2
ddm2 3 2 ddm2
c1 80 90 5.460E-12
re1 10 12 2.246E+02
re2 11 12 2.246E+02
iee 12 4 6.000E-06
re 12 0 3.333E+07
ce 12 0 1.579E-12
* intermediate
gcm 0 8 12 0 2.841E-11
ga 8 0 80 90 1.131E-04
r2 8 0 1.000E+05
c2 1 8 3.000E-11
qb 1 0 8 0 1.960E+02
* output
ro1 1 6 1.000E+02
ro2 1 0 9.000E+02
rc 17 0 1.063E-04
qc 0 17 6 0 9.408E+03
d1 1 17 ddm1
d2 17 1 ddm1
d3 6 13 ddm2
d4 14 6 ddm2
vc 7 13 1.785E+00
ve 14 4 1.785E+00
ip 7 4 3.740E-04
dsub 4 7 ddm2
* models
.model qm1 npn (is=8.000E-16 bf=7.500E+04)
.model qm2 npn (is=8.003E-16 bf=1.500E+05)
.model ddm1 d (is=1.179E-19)
.model ddm2 d (is=8.000E-16)
.ends LT1012
*
.subckt LT1024 3 2 7 4 6
x_LT1024 3 2 7 4 6 LT1012
.ends LT1024
*
* - - - - - fini LT1012 family * - - - - - [oamm vnl 8/89]
```

Figure 1.

References

- Boyle, G.R., Cohn, B.M., Pederson, D.O., Solomon, J.E., "Macromodeling of Integrated Circuit Operational Amplifiers," *IEEE Journal of Solid-State Circuits*, Vol. SC-9, #6, December 1974.
- Jung, W.G., "An LT1013 Op Amp Macromodel," Linear Technology Design Note Number 13, July 1988.

Obtaining This Macromodel

This model can be entered onto a given computer type simply by typing it in (very carefully!) using an ASCII text editor. Optionally, interested readers may contact LTC at the address or phone number below, for a copy of a PC data disc containing the most recent collection of macromodels (including this model and all those previously released).

For literature on the LT1012 op amp, or a 5 1/4 " macromodel disk, call (800) 637-5545. For applications help, call (408) 432-1900, Ext. 456.